// File: contracts/Minter.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.15;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/utils/cryptography/MerkleProof.sol";
import "@openzeppelin/contracts/utils/Address.sol"; 
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "./IToken.sol";

contract Minter is Ownable, ReentrancyGuard {
    using SafeERC20 for IERC20;

    // ======== Supply =========
    uint256 public maxMintsPerAddress;
    uint256 public maxTokens;

    // ======== Cost =========;
    uint256 public immutable tokenPriceETH;
    uint256 public tokenPriceETH_Discount;
    uint256 public tokenPriceApeCoin;
    uint256 public tokenPriceApeCoin_Discount;
    
    event TokenPriceApeCoinChanged(uint256 oldPrice, uint256 newPrice);
    event TokenPriceApeCoinDiscountChanged(uint256 oldPrice, uint256 newPrice);
    event TokenPriceETHDiscountChanged(uint256 oldPrice, uint256 newPrice);

    // ======== Sale Status =========
    bool public saleIsActive = false;
    bool public saleApeCoinIsActive = false;
    bool public isWhitelistSaleActive = false;
    bool public isDiscountedSaleActive = false;
    bool public isPublicSaleActive = false;

    // ======== Claim Tracking =========
    mapping(address => uint256) private addressToMintCount;
    mapping(address => bool) public whitelistClaimed;
    mapping(address => uint256) public discountClaimed;

    // ======== Whitelist Validation =========
    bytes32 public whitelistMerkleRoot;
    bytes32 public discountMerkleRoot;

    // ======== External Storage Contract =========
    IToken public immutable token;

    // ======== Fund Management =========
    // NFF Gnosis Wallet
    address public withdrawalAddress = 0xAEE6a9660878A217f4070181FFb271c4De61fBac;

    // ======== ApeCoin Support =========
    IERC20 public immutable tokenApeCoin;

    // ======== Constructor =========
    constructor(address contractAddress,
                uint256 _maxTokens,
                uint256 _maxMintsPerAddress,
                IERC20 _tokenApeCoin,
                uint256 _tokenPriceETH_Discount,
                uint256 _tokenPriceETH,
                uint256 _tokenPriceApeCoin_Discount,
                uint256 _tokenPriceApeCoin) {
        token = IToken(contractAddress);
        maxTokens = _maxTokens;
        maxMintsPerAddress = _maxMintsPerAddress;
        tokenApeCoin = _tokenApeCoin;
        tokenPriceETH = _tokenPriceETH;

        setTokenPriceETHDiscount(_tokenPriceETH_Discount);
        setTokenPriceApeCoin(_tokenPriceApeCoin);
        setTokenPriceApeCoinDiscount(_tokenPriceApeCoin_Discount);
    }

    // ======== Modifier Checks =========
    modifier isWhitelistMerkleRootSet() {
        require(whitelistMerkleRoot != 0, "Whitelist merkle root not set!");
        _;
    }

    modifier isValidMerkleProofWhitelist(address _address, bytes32[] calldata merkleProof, uint256 quantity) {
        require(
            MerkleProof.verify(
                merkleProof, 
                whitelistMerkleRoot, 
                keccak256(abi.encodePacked(keccak256(abi.encodePacked(_address, quantity)))
                )
            ), 
            "Address is not on whitelist!");
        _;

    }

    modifier isDiscountMerkleRootSet() {
        require(discountMerkleRoot != 0, "Discount merkle root not set!");
        _;
    }

    modifier isValidMerkleProofDiscount(address _address, bytes32[] calldata merkleProof, uint256 quantity) {
        require(
            MerkleProof.verify(
                merkleProof, 
                discountMerkleRoot, 
                keccak256(abi.encodePacked(keccak256(abi.encodePacked(_address, quantity)))
                )
            ), 
            "Address is not on discount list!");
        _;
    }
    
    modifier isSupplyAvailable(uint256 numberOfTokens) {
        uint256 supply = token.tokenCount();
        require(supply + numberOfTokens <= maxTokens, "Exceeds max token supply!");
        _;
    }

    modifier isSaleActive() {
        require(saleIsActive, "Sale is not active!");
        _;
    }

    modifier isMaxMintsPerAddressExceeded(uint amount) {
        require(addressToMintCount[msg.sender] + amount <= maxMintsPerAddress, "Exceeds max mint per address!");
        _;
    }


    /// @notice Set the discounted token price in ETH
    /// @param _tokenPriceETHDiscount The ETH price
    function setTokenPriceETHDiscount(uint256 _tokenPriceETHDiscount) public onlyOwner {
      emit TokenPriceETHDiscountChanged(tokenPriceETH_Discount, _tokenPriceETHDiscount);
      tokenPriceETH_Discount = _tokenPriceETHDiscount;
    }

    /// @notice Set the token price in ApeCoin
    /// @param _tokenPriceApeCoin The ApeCoin price
    function setTokenPriceApeCoin(uint256 _tokenPriceApeCoin) public onlyOwner {
      emit TokenPriceApeCoinChanged(tokenPriceApeCoin, _tokenPriceApeCoin);
      tokenPriceApeCoin = _tokenPriceApeCoin;
    }

    /// @notice Set the discounted token price in ApeCoin
    /// @param _tokenPriceApeCoinDiscount The ApeCoin price
    function setTokenPriceApeCoinDiscount(uint256 _tokenPriceApeCoinDiscount) public onlyOwner {
      emit TokenPriceApeCoinDiscountChanged(tokenPriceApeCoin_Discount, _tokenPriceApeCoinDiscount);
      tokenPriceApeCoin_Discount = _tokenPriceApeCoinDiscount;
    }

    /// @notice Transfer funds to gnosis wallet
    /// @param qty The qty user is purchsing
    /// @param applyDiscount Is this a discounted purchase
    function transferFunds(uint256 qty, bool applyDiscount) private {

        uint256 priceETH = !applyDiscount ? tokenPriceETH : tokenPriceETH_Discount;
        uint256 priceApeCoin = !applyDiscount ? tokenPriceApeCoin : tokenPriceApeCoin_Discount;

        if(msg.value == qty * priceETH) { 
            // pay with ETH
            (bool success, ) = payable(withdrawalAddress).call{value: qty * priceETH}("");
            require(success, "transfer failed");
        } else if(msg.value == 0) { 
            // pay with ApeCoin
            require(saleApeCoinIsActive, "ApeCoin sale is not active!");

            tokenApeCoin.safeTransferFrom(
              msg.sender,
              withdrawalAddress,
              qty * priceApeCoin
            );
        } else {
          revert("invalid payment option");
        }
    }

    // ======== Mint Functions =========
    /// @notice Mint all available tokens on whitelist
    /// @param merkleProof The merkle proof generated offchain
    /// @param amount The amount user can mint
    function mintWhitelist(bytes32[] calldata merkleProof, uint amount) public 
        isSaleActive()
        isWhitelistMerkleRootSet()
        isValidMerkleProofWhitelist(msg.sender, merkleProof, amount) 
        isSupplyAvailable(amount) 
        isMaxMintsPerAddressExceeded(amount)
        nonReentrant {
            require(isWhitelistSaleActive, "Whitlist sale is not active!");
            require(!whitelistClaimed[msg.sender], "Whitelist is already claimed by this wallet!");

            token.mint(amount, msg.sender);

            addressToMintCount[msg.sender] += amount;

            whitelistClaimed[msg.sender] = true;
    }

    /// @notice Mint tokens at a discounted price
    /// @param merkleProof The merkle proof generated offchain
    /// @param amount The amount user can mint
    /// @param quantity The quantity user would like to mint
    function mintDiscount(bytes32[] calldata merkleProof, uint amount, uint quantity) public payable 
        isSaleActive()
        isDiscountMerkleRootSet()
        isValidMerkleProofDiscount(msg.sender, merkleProof, amount) 
        isSupplyAvailable(quantity) 
        isMaxMintsPerAddressExceeded(quantity)
        nonReentrant {
            require(isDiscountedSaleActive, "Discount sale is not active!");
            require(discountClaimed[msg.sender] != amount, "Discount is already claimed by this wallet!");

            token.mint(quantity, msg.sender);

            transferFunds(quantity, true);

            addressToMintCount[msg.sender] += quantity;
            discountClaimed[msg.sender]  += quantity;
    }


    /// @notice Mint tokens at public price
    /// @param quantity The quantity user would like to mint
    function mintPublic(uint quantity) public payable 
        isSaleActive()
        isSupplyAvailable(quantity) 
        isMaxMintsPerAddressExceeded(quantity)
        nonReentrant  {
            require(isPublicSaleActive, "Public sale is not active!");
            require(msg.sender == tx.origin, "Mint: not allowed from contract");

            transferFunds(quantity, false);

            token.mint(quantity, msg.sender);

            addressToMintCount[msg.sender] += quantity;
    }

    /// @notice Mint team tokens at zero cost
    /// @param _to The address to send the tokens
    /// @param _reserveAmount The quantity user would like to mint
    function mintTeamTokens(address _to, uint256 _reserveAmount) public 
        onlyOwner 
        isSupplyAvailable(_reserveAmount) {
            token.mint(_reserveAmount, _to);
    }

    // ======== Whitelisting =========
    /// @notice Set merkle root for whitelist mints
    /// @param merkleRoot The merkle root
    function setWhitelistMerkleRoot(bytes32 merkleRoot) external onlyOwner {
        whitelistMerkleRoot = merkleRoot;
    }

    /// @notice Check if user is whitelisted
    /// @param _address The whitelisted address
    /// @param merkleProof The merkle proof generated offchain
    /// @param amount The number of tokens the user has been whitelisted for
    function isWhitelisted(address _address, bytes32[] calldata merkleProof, uint256 amount) external view
        isValidMerkleProofWhitelist(_address, merkleProof, amount) 
        returns (bool) {            
            require(!whitelistClaimed[_address], "Whitelist is already claimed by this wallet");

            return true;
    }

    /// @notice Check if user has claimed all their whitelist spots
    /// @param _address The whitelisted address
    function isWhitelistClaimed(address _address) external view returns (bool) {
        return whitelistClaimed[_address];
    }


    // ======== Discounting =========
    /// @notice Set merkle root for discounted mints
    /// @param merkleRoot The merkle root
    function setDiscountMerkleRoot(bytes32 merkleRoot) external onlyOwner {
        discountMerkleRoot = merkleRoot;
    }

    /// @notice Check if user is on discount list
    /// @param _address The discounted address
    /// @param merkleProof The merkle proof generated offchain
    /// @param amount The number of tokens the user has been whitelisted for
    function isDiscounted(address _address, bytes32[] calldata merkleProof, uint256 amount) external view
        isValidMerkleProofDiscount(_address, merkleProof, amount) 
        returns (bool) {            
            require(discountClaimed[_address] != amount, "Discount is already claimed by this wallet");

            return true;
    }

    /// @notice Returns the number of discounted spots user has claimed
    /// @param _address The discounted address
    function discountsClaimed(address _address) external view returns (uint256) {
        return discountClaimed[_address];
    }

    // ======== Utilities =========
    /// @notice Returns the number of tokens minted by an address
    /// @param _address The minter's address
    function mintCount(address _address) external view returns (uint) {
        return addressToMintCount[_address];
    }

    // ======== State Management =========
    /// @notice Toggle mint sale status
    function flipSaleStatus() public onlyOwner {
        saleIsActive = !saleIsActive;
    } 

    /// @notice Toggle whitelist mint sale status
    function flipWhitelistSaleStatus() public onlyOwner {
        isWhitelistSaleActive = !isWhitelistSaleActive;
    } 

    /// @notice Toggle discounted mint sale status
    function flipDiscountedSaleStatus() public onlyOwner {
        isDiscountedSaleActive = !isDiscountedSaleActive;
    }

    /// @notice Toggle public mint sale status
    function flipPublicSaleStatus() public onlyOwner {
        isPublicSaleActive = !isPublicSaleActive;
    } 

    /// @notice Toggle ApeCoin sale status
    function flipSaleApeCoinStatus() public onlyOwner {
        saleApeCoinIsActive = !saleApeCoinIsActive;
    }

    // ======== Token Supply Management=========
    /// @notice Set the max mints per address
    /// @param _max The maximum number of tokens a user can mint
    function setMaxMintPerAddress(uint _max) public onlyOwner {
        maxMintsPerAddress = _max;
    }

    /// @notice Set the max supply, can only be lowered not increased
    /// @param newMaxTokenSupply The maximum number of tokens that can be minted
    function decreaseTokenSupply(uint256 newMaxTokenSupply) external onlyOwner {
        require(maxTokens > newMaxTokenSupply, "Max token supply can only be decreased!");
        maxTokens = newMaxTokenSupply;
    }

    // ======== Withdraw =========
    /// @notice Set the withdrawal address for all funds
    /// @param _newWithdrawalAddress The withdrawal address for buth ETH/ApeCoin funds
    function setWithdrawalAddress(address _newWithdrawalAddress) public onlyOwner {
        withdrawalAddress = _newWithdrawalAddress;
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and make it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/utils/cryptography/MerkleProof.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev These functions deal with verification of Merkle Trees proofs.
 *
 * The proofs can be generated using the JavaScript library
 * https://github.com/miguelmota/merkletreejs[merkletreejs].
 * Note: the hashing algorithm should be keccak256 and pair sorting should be enabled.
 *
 * See `test/utils/cryptography/MerkleProof.test.js` for some examples.
 */
library MerkleProof {
    /**
     * @dev Returns true if a `leaf` can be proved to be a part of a Merkle tree
     * defined by `root`. For this, a `proof` must be provided, containing
     * sibling hashes on the branch from the leaf to the root of the tree. Each
     * pair of leaves and each pair of pre-images are assumed to be sorted.
     */
    function verify(
        bytes32[] memory proof,
        bytes32 root,
        bytes32 leaf
    ) internal pure returns (bool) {
        bytes32 computedHash = leaf;

        for (uint256 i = 0; i < proof.length; i++) {
            bytes32 proofElement = proof[i];

            if (computedHash <= proofElement) {
                // Hash(current computed hash + current element of the proof)
                computedHash = keccak256(abi.encodePacked(computedHash, proofElement));
            } else {
                // Hash(current element of the proof + current computed hash)
                computedHash = keccak256(abi.encodePacked(proofElement, computedHash));
            }
        }

        // Check if the computed hash (root) is equal to the provided root
        return computedHash == root;
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: contracts/IToken.sol



// Contract based on https://docs.openzeppelin.com/contracts/3.x/erc721
// SPDX-License-Identifier: MIT
/// @title Interface for Token

pragma solidity ^0.8.15;

abstract contract IToken {

    function setProvenanceHash(string memory _provenanceHash) virtual external;

    function mint(uint256 _count, address _recipient) virtual external;

    function setBaseURI(string memory baseURI) virtual external;

    function updateMinter(address _minter) virtual external;

    function lockMinter() virtual external;

    function tokenCount() virtual external returns (uint256);
}
// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}